

<u>Problem description:</u>

This example represents the propagation of a seismic wave in a square region. The wave is generated at the centre of the region, while a "receiver" (displacement transducer) is located at a certain height over the source.
The region immediately around the source is modelled by spectral elements, while the remaining parts of the domain may either be meshed by finite elements (hybrid solution), or by spectral elements (purely spectral solution).
By comparing the two solutions, an assessment of the quality of FE/SE coupling is obtained.

<u>Numerical Solutions</u>

HOLE06

This calculation assumes a hybrid mesh composed of 4 macro spectral elements, 64 micro spectral elements and 2400 finite elements.

The coupling between FE and SE regions along the closed square interface is realized by means of the LINK COUP FESE directive.

The mesh generation file is:

```
*%siz 50
*
opti echo 1;
opti titr 'HOLE - 06';
*
opti echo 0;
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxleg2.proc';
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxspect2.proc';
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxpdroi1.proc';
opti echo 1;
*
```

```
opti dime 2 elem qua4;
p0 = 0 0;
p1 = -250.0 -250.0;
p2 = 250.0 -250.0;
p3 = -50.0 -50.0;
p4 =  50.0 -50.0;
p5 = -50.0 50.0;
p6 =  50.0 50.0;
p7 = -250 50.0;
p8 = 250.0 50.0;
p9 = -250 -50.0;
```

```
p10 = 250.0 -50.0;
*
p31 = p3 plus p0;
p41 = p4 plus p0;
p51 = p5 plus p0;
p61 = p6 plus p0;
n = 4;
tol=0.001;
*
c1 = p1 d 50 p2;
s1 = c1 tran 20 (0 200);
* trac s1;
c2 = p4 d 20 p10;
s2 = c2 tran 10 (0 100);
* trac s2;
c3 = p9 d 20 p3;
s3 = c3 tran 10 (0 100);
* trac s3;
c4 = p7 d 50 p8;
s4 = c4 tran 20 (0 200);
* trac s4;
* opti donn 5;
ss = s2 et s3;
elim tol (s1 et ss et p3 et p4 et p5 et p6);
* q1 = diff s1 ss;
fem = diff s1 ss;
elim tol (fem et s4 et p7 et p8 et p5 et p6);
fem = fem et s4;
trac fem;
c5 = p31 d 2 p41;
s5 = c5 tran 2 (0 100);
* trac (fem et s5);
sem = pxspect2 s5 n tol;
elim tol (s5 et p51 et p61);
trac (fem et sem);
mesh = fem et sem et s5;
nodf = pxpdroi1 fem p3 p4 tol;
nodf = nodf et (pxpdroi1 fem p4 p6 tol);
nodf = nodf et (pxpdroi1 fem p3 p5 tol);
nodf = nodf et (pxpdroi1 fem p5 p6 tol);
nods = pxpdroi1 sem p31 p41 tol;
nods = nods et (pxpdroi1 sem p41 p61 tol);
nods = nods et (pxpdroi1 sem p31 p51 tol);
nods = nods et (pxpdroi1 sem p51 p61 tol);
*
tpln=0.0 100.0;
elim tol (tpln et fem);
*
mesh= mesh et nodf et nods;
tass mesh;
*
opti sauv form 'hole06.msh';
sauv form mesh;
*
opti donn 5;
fin;
```

The input file is:

```
HOLE - 06
$
ECHO
$VERI
!CONV win
CAST MESH
DPLA NONL
$
DIME
  PT2L 2601 MS24 4 S24 64 CAR1 2400 ZONE 3
TERM
$
GEOM MS24 s5 S24 sem CAR1 fem TERM
$
MATE LINE RO 2500. YOUN 1.501196172E9 NU 0.200956938
          LECT s5 fem TERM
$
LINK COUP
     FESE FNOD LECT nodf TERM
          SNOD LECT nods TERM
CHAR SPEC POIN BET DELT
               SOUR BETA 50.0
                    AMP 10.0 X 0.0 Y 0.0
                    T0 0.01 ALFA 40.0
                    NX 0 NY 1
$
ECRI DEPL VITE ACCE FINT FEXT TFRE 0.25
     FICH ALIC FREQ 1
$
OPTI PAS UTIL NOTE
 LOG 1
$
CALC TINI 0. TEND 0.5d0 PASF 5.0D-3
*==============================================================
SUIT
Post-treatment (bande alice)
ECHO
*
RESU alic GARD PSCR
*
SORT GRAP
*
AXTE 1.0 'Time [s]'
*
COUR 1 'dy_rec'  DEPL COMP 2 NOEU LECT tpln TERM
*
trac 1 AXES 1.0 'DISPL. [M]' yzer
list 1 AXES 1.0 'DISPL. [M]'
*
QUAL DEPL COMP 2 LECT tpln TERM REFE -5.50193E-11 TOLE 3.E-3
*==============================================================
$
FIN
```

HOLEPS

This calculation assumes a purely spectral mesh composed of 225 macro spectral elements and 3600 micro spectral elements.

The mesh generation file is:

```
*%siz 50
*
opti echo 1;
opti titr 'HOLE - PS';
*
opti echo 0;
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxleg2.proc';
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxspect2.proc';
opti echo 1;
*
opti dime 2 elem qua4;
p0 = 0 0;
p1 = -250.0 -250.0;
p2 = 250.0 -250.0;
n = 4;
tol=0.001;
*
c1 = p1 d 15 p2;
sm = c1 tran 15 (0 500);
trac sm;
elim tol (sm et p1 et p2);
sem = pxspect2 sm n tol;
trac (sm et sem);
*
tpln=0.0 100.0;
elim tol (tpln et sem);
*
mesh = sm et sem ;
*
tass mesh;
*
opti sauv form 'holePS.msh';
sauv form mesh;
*
opti donn 5;
fin;
```

The input file is:

```
HOLE - PS
$
ECHO
$VERI
!CONV win
CAST MESH
DPLA NONL
$
DIME
  PT2L 3721 MS24 225 S24 3600 ZONE 2
TERM
$
GEOM MS24 sm S24 sem TERM
$
MATE LINE RO 2500. YOUN 1.501196172E9 NU 0.200956938
          LECT sm TERM
$
CHAR SPEC POIN BET DELT
               SOUR BETA 50.0
                    AMP 10.0 X 0.0 Y 0.0
                    T0 0.01 ALFA 40.0
                    NX 0 NY 1
$
ECRI DEPL VITE ACCE FINT FEXT TFRE 0.25
     FICH ALIC FREQ 2
$
OPTI PAS UTIL NOTE
 LOG 1
$
CALC TINI 0. TEND 0.5d0 PASF 2.5D-3
*==============================================================
SUIT
Post-treatment (bande alice)
ECHO
*
RESU alic GARD PSCR
*
SORT GRAP
*
AXTE 1.0 'Time [s]'
*
COUR 1 'dy_rec_ps' DEPL COMP 2 NOEU LECT tpln TERM
RCOU 2 'dy_rec' FICH 'hole06.pun'
*
trac 1 AXES 1.0 'DISPL. [M]' yzer
trac 2 1 AXES 1.0 'DISPL. [M]' yzer
         COLO noir roug
*
QUAL DEPL COMP 2 LECT tpln TERM REFE  1.67558E-11 TOLE 3.E-3
*==============================================================
FIN
```

The receiver signal in the coupled FE/SE case is:



The receiver signal in the purely spectral case is:



The final X-displacement in the hybrid and purely spectral solutions are:


HOLE - 06
Time: 5.00000E-01 Step: 100
MAX:+9.01E-10
---:-5.50E-10
---:-6.50E-10
MIN:-9.01E-10
Displacement comp _X [m]
HOLE - PS
Time: 5.00000E-01 Step: 200
MAX:+8.84E-10
---:-5.50E-10
---:-6.50E-10
MIN:-8.84E-10
Displacement comp _X [m]